

## 5: BASIC Stamp Command Reference – FREQOUT

### FREQOUT

BS1	BS2	BS2e	BS2sx	BS2p	BS2pe	BS2px
-----	-----	------	-------	------	-------	-------



(See SOUND)



**FREQOUT** *Pin, Duration, Freq1 { , Freq2 }*

### Function

Generate one or two sine-wave tones for a specified *Duration*.

- **Pin** is a variable/constant/expression (0 – 15) that specifies the I/O pin to use. This pin will be set to output mode.
- **Duration** is a variable/constant/expression (0 - 65535) specifying the amount of time to generate the tone(s). The unit of time for *Duration* is described in Table 5.26.
- **Freq1** is a variable/constant/expression (0 – 32767) specifying frequency of the first tone. The unit of *Freq1* is described in Table 5.26.
- **Freq2** is an optional argument exactly like *Freq1*. When specified, two frequencies will be mixed together on the specified I/O pin.

### Quick Facts

Table 5.26: FREQOUT Quick Facts.

	BS2, BS2e	BS2sx	BS2p	BS2pe	BS2px
Units in <i>Duration</i>	1 ms	0.4 ms	0.265 ms	1 ms	0.166 ms
Units in <i>Freq1</i> and <i>Freq2</i>	1 Hz	2.5 Hz	3.77 Hz	1.51 Hz	6.03 Hz
Range of Frequency	0 to 32767 Hz	0 to 81917 Hz	0 to 123531 Hz	0 to 49478 Hz	0 to 197585 Hz
Related Commands	DTMFOUT and PWM				

### Explanation

FREQOUT generates one or two sine waves using a pulse-width modulation algorithm. The circuits shown in Figure 5.6 will filter the signal in order to play the tones through a speaker or audio amplifier. Here's a simple FREQOUT command:

SIMPLEST FORM OF FREQOUT.

```
FREQOUT 2, 1000, 2500
```

## FREQOUT – BASIC Stamp Command Reference

On the BS2, this command generates a 2500 Hz tone for 1 second (1000 ms) on I/O pin 2. See Table 5.26 for timing data on other BASIC Stamp models.


To play two tones on the same I/O pin at once:

```
FREQOUT 2, 1000, 2500, 3000
```

GENERATING TWO TONES AT ONCE.


This will generate a 2500 Hz and 3000 Hz tone for 1 second. The frequencies will mix together for a chord- or bell-like sound. To generate a silent pause, specify frequency value(s) of 0.

### Driving an Audio Amplifier



**Figure 5.6:** Example RC filter circuits for driving an audio amplifier(top) or a speaker (bottom).

### Driving a Speaker



Notes:  
C1 may be omitted for piezo speakers  
C2 is optional, but reduces high-frequency noise

The circuits in Figure 5.6 work by filtering out the high-frequency PWM used to generate the sine waves. FREQOUT works over a very wide range of frequencies (as shown in Table 5.26) so at the upper end of its range, those PWM filters will also filter out most of the desired frequency. You may find it necessary to reduce values of the parallel capacitors shown in the circuit, or to devise a custom active filter for your application.

FREQUENCY CONSIDERATIONS.

## 5: BASIC Stamp Command Reference – FREQOUT



NOTE: This example program can be used with all BS2 models. This program uses conditional compilation techniques; see Chapter 3 for more information.

### Demo Program (FREQOUT.bs2)

```
' FREQOUT.bs2
' This program demonstrates sound-effects generation by the BASIC Stamp.
' Conditional compilation sets timing and frequency adjustment factors so
' that the output will sound the same on any BS2 model.

' {$STAMP BS2}
' {$PBASIC 2.5}

Spkr          PIN      10          ' output pin for FREQOUT

#SELECT $STAMP
#CASE BS2, BS2E
  TmAdj        CON      $100        ' x 1.0 (time adjust)
  FrAdj        CON      $100        ' x 1.0 (freq adjust)
#CASE BS2SX
  TmAdj        CON      $280        ' x 2.5
  FrAdj        CON      $066        ' x 0.4
#CASE BS2P
  TmAdj        CON      $3C5        ' x 3.77
  FrAdj        CON      $044        ' x 0.265
#CASE BS2PE
  TmAdj        CON      $100        ' x 1.0
  FrAdj        CON      $0A9        ' x 0.662
#CASE BS2PX
  TmAdj        CON      $607        ' x 6.03
  FrAdj        CON      $2A         ' x 0.166
#ENDSELECT

Main:
  DEBUG "Let's make a call...", CR
  ' combine 350 Hz & 440 Hz
  FREQOUT Spkr, 2000 */ TmAdj, 350 */ FrAdj, 440 */ FrAdj
  ' dial number (digits 150 ms on, 25 ms off)
  DTMFOUT Spkr, 150 */ TmAdj, 25, [5, 5, 5, 1, 2, 1, 2]
  PAUSE 500

  ' bad connection (SIT sequence)
  FREQOUT Spkr, 375 */ TmAdj, 985 */ FrAdj
  FREQOUT Spkr, 375 */ TmAdj, 1371 */ FrAdj
  FREQOUT Spkr, 375 */ TmAdj, 1777 */ FrAdj

  DEBUG "Oops! -- try again...", CR
  PAUSE 1000
  DTMFOUT Spkr, 150 */ TmAdj, 25, [5, 5, 5, 2, 2, 2, 2]
  DEBUG "Ringing"
  FREQOUT Spkr, 2000 */ TmAdj, 440 */ FrAdj, 480 */ FrAdj
  PAUSE 4000
  FREQOUT Spkr, 2000 */ TmAdj, 440 */ FrAdj, 480 */ FrAdj
  INPUT Spkr
  END
```

## ***FREQOUT – BASIC Stamp Command Reference***

---